National Bureau of Standards
Statistical Engineering Divisision
Gaithersburg, MD 20899

N-86-1
October 1986

Accurate Computation of the Log of the Gamma Function

Charles P. Reeve

The gamma function, $\Gamma(x) = \int_0^\infty t^{x-1}e^{-t}dt$ $(x>0)$, appears as a constant in several commonly used statistical distribution and density functions. For example, the p.d.f. of the gamma distribution is

$$f(t) = t^{\nu-1}e^{-t}/\Gamma(\nu) \quad (\nu>0).$$

If one tried to evaluate this function directly on a computer when, say ν=228 and t=200, it is likely that the components 200^{227} and $\Gamma(228)$ would overflow while e^{-200} might underflow. However, if f(t) is expressed as

$$f(t) = e^{(\nu-1)\ln t - t - \ln\Gamma(\nu)}$$

then, using 8 digit arithmetic,

$$\begin{aligned} f(200) &= e^{227\ln 200 - 200 - \ln\Gamma(228)} \\ &= e^{1202.7180 - 200 - 1008.0954} \\ &= e^{-5.3774} \\ &= 0.0046198. \end{aligned} \tag{1}$$

The overflow/underflow problem has been circumvented, but at the expense of three digits of accuracy. As a final result this might be acceptable, but as an intermediate step in a larger calculation the dropoff in accuracy could be disastrous.

Consider the computation e^τ where τ is a real number. Let $T=\tau+\delta$ be the computer representation of τ where δ is the <u>absolute</u> roundoff error. The <u>relative</u> error in e^T is then $(e^T-e^\tau)/e^\tau = (e^{\tau+\delta}-e^\tau)/e^\tau = e^\delta-1 \cong 1+\delta-1 = \delta$ when $|\delta|<<1$. Returning to (1), the absolute error in the exponent is on the order of 10^{-5} so, as predicted, the relative error in the result is also about 10^{-5}.

For all practice purposes the "loss of accuracy" problem in (1) can be solved by doing the computations in double precision. Suppose that T_1, T_2, and T_3 are double precision variables in a computer which carries 8 digits in single precision and 16 digits in double precision. If $|T_i|<10^8$ (i=1,2,3) then each T_i will carry at least 8 correct digits to the right of the decimal as will $T_1+T_2+T_3$. The result $e^{T_1+T_2+T_3}$ will then be correct to the single precision limit of 8 digits as shown above. If one of the T_i is $\ln\Gamma(\nu)$ then the constraint $|\ln\Gamma(\nu)|<10^8$ will be satisfied when $e^{-10^8} < \nu < 6{,}788{,}524$. This surely covers all ν of practical interest.

When the above computations are implemented on a computer it is necessary to have a good algorithm for computing $\ln\Gamma(x)$ in double precision for $x>0$. Many software packages [2,3,4] contain such algorithms. However, the user may not know with confidence their absolute accuracy or may not be able to get hold of their source code for implementation on a computer not having one of these packages. A simple algorithm is given below for computing $\ln\Gamma(x)$ in double precision to a pre-set absolute accuracy.

In equation 6.1.40 of Abramowitz and Stegun [1] an infinite series representation for $\ln\Gamma(x)$ is given which can be expressed as

$$\ln\Gamma(x) = (x - 1/2)\ln x - x - \ln(2\pi)/2 + \sum_{m=1}^{N} T_m + R_N \qquad (2)$$

where $T_m = \dfrac{B_{2m}}{2m(2m-1)x^{2m-1}}$, R_N is the remainder, and the B_{2m} are Bernoulli numbers. Values of B_{2m} and $B_{2m}/[2m(2m-1)]$ (m=1,2,...,9) are given in table 1. Whittaker and Watson [5] show that $|R_N|<|T_{N+1}|$.

This algorithm has been implemented in the FORTRAN double precision function GAMLOG with N=8, thus the bound on the remainder is

$$\begin{aligned} |R_8| &< |T_9| \\ &< B_{18}/[(18)(17)x^{17}] \\ &< 43867/(244188x^{17}). \end{aligned} \qquad (3)$$

In table 2 minimum values of x, denoted x_{min}, have been computed which satisfy (3) for certain values of $|T_9|$. On the CDC Cyber 180/855 and Cyber 205 at NBS fifteen digit accuracy is sufficient for single precision computations, thus a pre-set value of XMIN = 6.894 is currently in GAMLOG. When $0<x<x_{min}$ accuracy is maintained using the relationship $\Gamma(x) = \Gamma(x+1)/x$. For example, $\ln\Gamma(3.7)$ is computed by first computing $\ln\Gamma(7.7)$ by (2), without the R_N term, then setting $\ln\Gamma(3.7) = \ln\Gamma(7.7) - \ln[(3.7)(4.7)(5.7)(6.7)]$. The program accepts only positive values of x.

Before returning the value of $\ln\Gamma(x)$ GAMLOG checks to see whether it is too large to meet the absolute accuracy criterion (ABSACC = 1.0E-15). This is accomplished by testing whether $\ln\Gamma(x) + 10^{-15} = \ln\Gamma(x)$. If it is then the absolute accuracy criterion has not been achieved and an error message is printed.

A listing of GAMLOG is an appendix to this note. It is the user's responsibility to see that the pre-set variables XMIN and ABSACC are appropriate for the computer being used.

Table 1

Bernoulli numbers (B_{2m}) and coefficients of $1/x^{2m-1}$ in the truncated series representation of $\ln\Gamma(x)$ in (2).

m	B_{2m}	$B_{2m}/[(2m)(2m-1)]$
1	1 / 6	1 / 12
2	-1 / 30	-1 / 360
3	1 / 42	1 / 1260
4	-1 / 30	-1 / 1680
5	5 / 66	1 / 1188
6	-691 / 2730	-691 / 360360
7	7 / 6	1 / 156
8	-3617 / 510	-3617 / 122400
9	43867 / 798	43867 / 244188

Table 2

Minimum values of x for which the truncated series representation of $\ln\Gamma(x)$ in (2) has an absolute accuracy bounded by $|T_9| = 10^{-d}$.

d	x_{min}	d	x_{min}	d	x_{min}
1	1.035	11	4.011	21	15.539
2	1.185	12	4.592	22	17.793
3	1.357	13	5.258	23	20.374
4	1.554	14	6.021	24	23.330
5	1.779	15	6.894	25	26.713
6	2.037	16	7.894	26	30.588
7	2.333	17	9.039	27	35.025
8	2.671	18	10.351	28	40.105
9	3.059	19	11.852	29	45.922
10	3.502	20	13.571	30	52.583

References

1. Abramowitz, Milton and Stegun, Irene A., Handbook of Mathematical Functions, NBS Applied Mathematics Series 55, 1970, p. 257.
2. IMSL, Inc., Houston, TX. [MLGAMD]
3. NBS Core Math Library (CMLIB). [DLNGAM]
4. Numerical Algorithms Group (NAG), Downers Grove, IL. [S14ABF]
5. Whittaker, E.T. and Watson, G.N., A Course of Modern Analysis, The MacMillan Company, 1947, pp. 251-3.

```
      DOUBLE PRECISION FUNCTION GAMLOG (X)
C
C-----------------------------------------------------------------------
C   GAMLOG   WRITTEN BY CHARLES P. REEVE, STATISTICAL ENGINEERING
C            DIVISION, NATIONAL BUREAU OF STANDARDS, GAITHERSBURG,
C            MARYLAND  20899
C
C   FOR: COMPUTING THE DOUBLE PRECISION LOG OF THE GAMMA FUNCTION WITH
C        SINGLE PRECISION PARAMETER X>0.  THE MAXIMUM TRUNCATION ERROR
C        IN THE INFINITE SERIES (SEE REFERENCE 1) IS DETERMINED BY THE
C        CONSTANT XMIN.  WHEN X<XMIN A RECURRENCE RELATION IS USED IN
C        ORDER TO ACHIEVE THE REQUIRED ABSOLUTE ACCURACY.  THE TABLE
C        BELOW GIVES THE MINIMUM VALUE OF X WHICH YIELDS THE CORRES-
C        PONDING ABSOLUTE ACCURACY IN GAMLOG(X) ASSUMING THE MACHINE
C        CARRIES ENOUGH DIGITS WHEN THOSE TO THE LEFT OF THE DECIMAL
C        ARE CONSIDERED (SEE REFERENCE 2 FOR FURTHER DISCUSSION).  IF
C        THE LATTER CONDITION IS NOT MET, AN ERROR MESSAGE IS PRINTED.
C
C        THE CYBER 180/855 AT NBS CARRIES ABOUT 15 DIGITS IN SINGLE
C        PRECISION, THEREFORE THE PRE-SET VALUE OF ABSACC IS 10**(-15)
C        AND THE CORRESPONDING VALUE OF XMIN IS 6.894.  ON A DIFFERENT
C        MACHINE THESE CONSTANTS SHOULD BE CHANGED ACCORDINGLY.
C
C           XMIN    ACCURACY     XMIN    ACCURACY     XMIN    ACCURACY
C          ------   --------    ------   --------    ------   --------
C           1.357     1E-3       4.592     1E-12     15.539     1E-21
C           2.037     1E-6       6.894     1E-15     23.330     1E-24
C           3.059     1E-9      10.351     1E-18     35.025     1E-27
C
C   SUBPROGRAMS CALLED: -NONE-
C
C   CURRENT VERSION COMPLETED OCTOBER 25, 1986
C
C   REFERENCES:
C
C   1) ABRAMOWITZ, MILTON AND STEGUN, IRENE, 'HANDBOOK OF MATHEMATICAL
C      FUNCTIONS', NBS APPLIED MATHEMATICS SERIES 55, NOV. 1970,
C      EQ. 6.1.40, P 257.
C
C   2) REEVE, CHARLES P., 'ACCURATE COMPUTATION OF THE LOG OF THE GAMMA
C      FUNCTION', STATISTICAL ENGINEERING DIVISION NOTE 86-1, OCTOBER
C      1986.
C-----------------------------------------------------------------------
C
      DOUBLE PRECISION ABSACC,B1,B2,B3,B4,B5,B6,B7,B8,C,DX,Q,R,XMIN,XN
      DATA XMIN,ABSACC / 6.894D0,1.0D-15 /
      DATA C / 0.918938533204672741780329736D0 /
      DATA B1 / 0.833333333333333333333333333D-1 /
      DATA B2 / -0.277777777777777777777777778D-2 /
      DATA B3 / 0.793650793650793650793650794D-3 /
      DATA B4 / -0.595238095238095238095238095D-3 /
      DATA B5 / 0.841750841750841750841750842D-3 /
      DATA B6 / -0.191752691752691752691752692D-2 /
```

```
      DATA B7 / 0.641025641025641025641025641D-2 /
      DATA B8 / -0.295506535947712418300653595D-1 /
C
C--- TERMINATE EXECUTION IF X<=0.0
C
      IF (X.LE.0.0) STOP  '*** X<=0.0 IN FUNCTION GAMLOG ***'
      DX = DBLE(X)
      N = MAX0(0,INT(XMIN-DX+1.0D0))
      XN = DX+DBLE(N)
      R = 1.0D0/XN
      Q = R*R
      GAMLOG = R*(B1+Q*(B2+Q*(B3+Q*(B4+Q*(B5+Q*(B6+Q*(B7+Q*B8)))))))+C+
     *    (XN-0.5D0)*DLOG(XN)-XN
C
C--- USE RECURRENCE RELATION WHEN N>0 (X<XMIN)
C
      IF (N.GT.0) THEN
         Q = 1.0D0
         DO 20 I   =   0, N-1
            Q = Q*(DX+DBLE(I))
   20    CONTINUE
         GAMLOG=GAMLOG-DLOG(Q)
      ENDIF
C
C--- PRINT WARNING IF ABSOLUTE ACCURACY HAS NOT BEEN ATTAINED
C
      IF (GAMLOG+ABSACC.EQ.GAMLOG) THEN
         PRINT *,' ********* WARNING FROM FUNCTION GAMLOG *********'
         PRINT *,' REQUIRED ABSOLUTE ACCURACY NOT ATTAINED FOR X = ',X
      ENDIF
      RETURN
      END
```